DETERMINATION OF THE THICKNESS-SHEAR RESONANCE FREQUENCY FOR PIEZOELECTRIC ACCELEROMETERS USING THE KLM MODEL

Mihai Ursu*

Department of Mechanics and Computer Programming, Technical University of Cluj-Napoca
Muncii Bvd. 103-105, 400641 Cluj-Napoca, Romania
mihai_ursu_1492@yahoo.com

Abstract

In this paper a complete equivalent electric circuit for a planar shear piezoelectric accelerometer is presented. Starting from the KLM (Krimholtz-Leedom-Matthaei) approach for a single piezoceramic plate sandwiched between two cladding layers, the model is extended in order to facilitate the frequency domain analysis for arbitrary geometries of the seismic masses. An original and complete PSpice circuit source is given for illustrating the spectra calculations.

INTRODUCTION

Most of the commercially available accelerometers have as sensing devices piezoelectric ceramics. Such transducers are more and more attractive due to their size, mass, electrical properties and price. They present different constructive geometries, depending on the positioning of the piezoceramic element relative to the seismic masses of the system and to the base. The thickness-mode shear type effect is considered in this work, for rectangular plates as illustrated in figure 1b. An exact equivalent circuit that separates the piezoelectric material into an electrical port and two acoustic ports, through the use of an ideal electromechanical transformer, is presented in figure 1a. The difficulties in deriving the analytical solutions for the wave equations are overcome by the use of electrical network and transmission-line theory. This model, commonly referred to as the KLM model [1],[6], can be used for analyzing free and mass loaded resonators, transient responses, for material coefficient determination, for frequency design of multilayer and array transducers.

Because of the immediate possibility of behavioral simulation, numerical plots for various analyses can be obtained with dedicated electrical engineering programs

such as SPICE (Simulation Program with Integrated Circuit Emphasis) [13],[14]. This paper presents an electrical-equivalent implementation for frequency analysis of such transducers. The mechanical components from the left and right sides of the piezoceramic plate have an electrical correspondent in the acoustical loads Z_L and Z_R, respectively. The KLM model uses the transmission line model which describes both the piezoelectric transformation between electrical and mechanical vibration, and the propagation of acoustic waves in analogy to the electrical waves. The transmission line has two ports where acoustical loads are applied, for an accelerometer these being the seismic mass at one side and the center base at the other side. This acoustical side represented by the transmission line is coupled to the electrical side by a transformer with the ratio 1:Φ. Thus, the influence of acoustical load variations on the transducer's electrical impedance can be modeled.



Figure 1 - The KLM model for a single piezoceramic plate

The electrical components and the acoustical loads from figure 1a are given by the following expressions [6],[8],[9]:

$$C = \varepsilon\, A/b \qquad Z_0 = \rho\, v^D A$$
$$\Phi = \frac{\omega\, Z_0}{2\, h_{15} \sin\left(\omega b / 2 v^D\right)} \qquad X = \frac{h_{15}^2}{\omega^2\, Z_0} \sin\left(\omega b / v^D\right) \tag{1}$$

$$Z_L = j\, \rho_L v_L A_L \tan\left(\omega b_L / v_L\right) \qquad Z_R = j\, \rho_R v_R A_R \tan\left(\omega b_R / v_R\right) \tag{2}$$

where C is the capacity, ε is the permittivity, A is the lateral area of the plates, b is the thickness of the piezoceramic plate, Z_0 is the characteristic acoustic impedance, ρ is the piezoceramic material density, v^D is the shear speed in the piezoceramic, Φ is the ideal transformer ratio, ω is the angular frequency, h_{15} is the shear piezoelectric pressure coefficient, X is the complex reactance, and the variables with the L and R

subscripts have the analogous meaning belonging to the left and right acoustical loads, respectively. The electrical components whose expressions are given by (1) and (2) will be used in the extended equivalent circuit from the following section. Because of their angular frequency dependency, an accurate circuit implementation for the frequency domain has to be done by using the Laplace type of controlled sources in SPICE analysis [4],[13].

ACCELEROMETER EQUIVALENT CIRCUIT APPROACH



Figure 2 - Planar shear accelerometer (a) top view, and (b) symmetrical geometry structure (exaggerated scale) illustrating two piezoceramic plates sandwiched between the same center base and identical seismic masses

The geometry of an accelerometer based on the thickness-mode shear-type operating principle is given in figure 2 [5],[7]. The seismic masses have a profile which fits inside of a preloading ring of radius R which confers mechanical rigidity to the system starting from its center-base. For the calculations presented in this work, the piezoceramic elements consist of PZ-23 [15], while the base is of stainless steel and the seismic masses are of titanium. The electrical connections are sketched in figure 2b. The ratio between the thickness:width:length for the piezoceramic plates is taken 1:4:10 in order to minimize the transverse vibrations influences [2],[10],[11],[12], and is kept constant during all calculations. Using the notations from figure 2a, the dimension Q is given by

$$Q = R - b - d_M/2 \tag{3}$$

where d_M is the thickness of the center base. The seismic masses must be represented by a large enough succession of thin layers in order to approximate their real shape which has contact with the preloading ring. If n is the number of seismic mass layers at one side, the width of the i^{th} layer is

$$l_i = 2\sqrt{R^2 - \left[b + d_M/2 + (i-1)Q/n\right]^2} \qquad i = \overline{n,1} \tag{4}$$

When the acoustical loads are made by multiple layers, the KLM approach has to use the transfer-matrix formalism. The model does not have any restriction regarding the number of layers, their thicknesses or their mechanical properties. The number *n* has to be chosen as a compromise between computational time and desired accuracy, which will stop increasing significantly from a value of *n* that can be determined and kept constant for a given geometry. In figure 3 is illustrated how the succession of layers can be used to form the new acoustical load for a specific seismic mass profile. The new expressions for calculating the load impedances, using each layer's characteristic acoustical impedance, are given by [1],[6]



Figure 3 - Transfer matrix approach for the acoustic loads

$$Z_{ci} = \rho_{LR} v_{LR} H l_i = \rho_{LR} v_{LR} A_i \tag{5}$$

$$Z_i = Z_{ci} \frac{Z_{i+1} + j Z_{ci} \tan\left(\omega Q/n v_{LR}\right)}{Z_{ci} + j Z_{i+1} \tan\left(\omega Q/n v_{LR}\right)} \qquad i = \overline{n-1,1} \tag{6}$$

$$Z_n = j Z_{cn} \tan\left(\omega l_n / v_{LR}\right) \qquad Z_L = Z_R = Z_1 \tag{7}$$

whre *H* is the length of all layers and *Q*/*n* is their identical thickness.

The SPICE subcircuit implementation for the acoustical loads is presented in figure 4. The source listing, where the Laplace type of controlled voltage source is used, is given by (8).



Figure 4 - Subcircuit implementation for acoustical loads

$$Z(f) = U(f)/I$$

```
.subckt KLM_LR 1 2
v1 3 2 ac 1
ez1 1 3 laplace {i(v1)}={Z_1()}
.ends KLM_LR
```

(8)

A similar transformation for the complex frequency domain is done also for the reactance X from figure 1a, whose expression is given in (1), yielding the new form

$$U_X(f) = I\, j\, X(f) \qquad \Rightarrow \qquad j\, X(s) = \frac{\sqrt{-1}\, h_{15}^2}{-s^2 Z_0} \sin\left(\sqrt{-s^2}\, b / v^D\right) \tag{9}$$

After applying all necessary transformations for obtaining the equivalent electric model for the accelerometer from figure 2, the complete circuit was obtained as presented in figure 5. Its correspondent PSpice source listing is given in figure 6, the main file KLM_planar.cir using the library file KLM.lib. Such a code can be tested with the evaluation version of PSpice version 10.0 [14] or any other SPICE simulator which can interpret the Laplace type of controlled sources.

The vibration induced by the base from figure 2b is represented by an independent voltage source vBase and a controlled voltage source eBase for the acoustical impedance. The two seismic masses which belong to the mechanical system are connected in the electrical circuit at the node 1, this being the point relative to which the transducer's output impedance Z_{KLM} is calculated. The reactances are implemented with two voltage sources, one independent for setting the current sense, and another one, dependent, for their complex value given by (9). The ideal transformers are represented by the dependent voltage sources eT and current sources gT.



Figure 5 - Complete equivalent KLM circuit for the planar shear accelerometer

```
*KLM.LIB
.param PI={4*atan(1)}
.param alpha={1/3}
*PZ-23
.param h15=1.42e9
.param vD=2307.5
.param ro=7700
.param eps11=1370
.param b=1e-3
.step param b 1e-3 5e-3 1e-3
.param A={10*b*4*b}
*Titanium - Seismic mass
.param roLR=4450
.param vLR=3120
.param VolLR={20*PI*pwr(b,3)/pwr(alpha,2)}
.param HLR={10*b}
.param ALR={A/alpha}
.param dLRguess={VolLR/ALR}
*Stainless Steel - Middle
.param roM=7890
.param vM=3090
.param AM={A/alpha}
.param dM={roLR/roM * ALR/AM * dLRguess}
.param R={2*b/alpha}
.param Q={R-b-dM/2}
.param n=10
.func l_func(i) { 2*sqrt(R*R-pwr(b+dM/2+(i-1)*
+Q/n,2))}
.func Zc(i) { roLR*vLR*HLR*l_func(i) }
.func Z_10() {sqrt(-1)*Zc(10)*tan(sqrt(-s*s)*
+Q/n/vLR)}
.func Z_9() {Zc(9)*
+(Z_10()+sqrt(-1)*Zc(9)*tan(sqrt(-s*s)*Q/n/vLR))/
+(Zc(9)+sqrt(-1)*Z_10()*tan(sqrt(-s*s)*Q/n/vLR))}
.func Z_8() {Zc(8)*
+(Z_9()+sqrt(-1)*Zc(8)*tan(sqrt(-s*s)*Q/n/vLR))/
+(Zc(8)+sqrt(-1)*Z_9()*tan(sqrt(-s*s)*Q/n/vLR))}
.func Z_7() {Zc(7)*
+(Z_8()+sqrt(-1)*Zc(7)*tan(sqrt(-s*s)*Q/n/vLR))/
+(Zc(7)+sqrt(-1)*Z_8()*tan(sqrt(-s*s)*Q/n/vLR))}
.func Z_6() {Zc(6)*
+(Z_7()+sqrt(-1)*Zc(6)*tan(sqrt(-s*s)*Q/n/vLR))/
+(Zc(6)+sqrt(-1)*Z_7()*tan(sqrt(-s*s)*Q/n/vLR))}
.func Z_5() {Zc(5)*
+(Z_6()+sqrt(-1)*Zc(5)*tan(sqrt(-s*s)*Q/n/vLR))/
+(Zc(5)+sqrt(-1)*Z_6()*tan(sqrt(-s*s)*Q/n/vLR))}
.func Z_4() {Zc(4)*
+(Z_5()+sqrt(-1)*Zc(4)*tan(sqrt(-s*s)*Q/n/vLR))/
+(Zc(4)+sqrt(-1)*Z_5()*tan(sqrt(-s*s)*Q/n/vLR))}
.func Z_3() {Zc(3)*
+(Z_4()+sqrt(-1)*Zc(3)*tan(sqrt(-s*s)*Q/n/vLR))/
+(Zc(3)+sqrt(-1)*Z_4()*tan(sqrt(-s*s)*Q/n/vLR))}
.func Z_2() {Zc(2)*
+(Z_3()+sqrt(-1)*Zc(2)*tan(sqrt(-s*s)*Q/n/vLR))/
+(Zc(2)+sqrt(-1)*Z_3()*tan(sqrt(-s*s)*Q/n/vLR))}
.func Z_1() {Zc(1)*
+(Z_2()+sqrt(-1)*Zc(1)*tan(sqrt(-s*s)*Q/n/vLR))/
+(Zc(1)+sqrt(-1)*Z_2()*tan(sqrt(-s*s)*Q/n/vLR))}
.subckt KLM_LR 1 2
v1 3 2 ac 1
ez1 1 3 laplace {i(v1)}={Z_1()}
.ends KLM_LR

*************************************

*KLM_planar.cir
.include klm.lib
.param eps0=8.8541e-12
.param Z_0={ro*vD*A}
.param cap={eps0*eps11*A/b}
.param tau_delay={b/vD}
RLgnd 11 0 1e12
RLMgnd 13 0 1e12
RMgnd 16 0 1e12
RRMgnd 18 0 1e12
RRgnd 20 0 1e12
XL 11 10 KLM_LR
TLL 10 11 12 13 Z0={Z_0} td={tau_delay}
TLR 12 13 14 16 Z0={Z_0} td={tau_delay}
vBase 15 14 ac 1
eBase 16 15 laplace {i(vBase)}=
+{sqrt(-1)*roM*vM*AM*tan(sqrt(-s*s)*dM/2/vM)}
TRL 14 16 17 18 Z0={Z_0} td={tau_delay}
TRR 17 18 19 20 Z0={Z_0} td={tau_delay}
XR 20 19 KLM_LR
vKLM 1 0 ac 0
CL  1 2 {cap}
RcapL 1 2 1e12
vXL 3 2 ac 0
eXL 4 3 laplace {i(vXL)}=
+{sqrt(-1)*h15*h15/(-s*s*Z_0)*sin(b*sqrt(-s*s)/vD)}
eTL 4 5 laplace {v(12)-v(13)}=
+{sqrt(-s*s)*Z_0/2/h15/sin(sqrt(-s*s)*b/2/vD)}
gTL 5 0 laplace {(v(12)-v(13))/Z_0}=
+{-2*h15*sin(sqrt(-s*s)*b/2/vD)/sqrt(-s*s)/Z_0}
CR  1 6 {cap}
RcapR 1 6 1e12
vXR 7 6 ac 0
eXR 8 7 laplace {i(vXR)}=
+{sqrt(-1)*h15*h15/(-s*s*Z_0)*sin(b*sqrt(-s*s)/vD)}
eTR 8 9 laplace {v(17)-v(18)}=
+{sqrt(-s*s)*Z_0/2/h15/sin(sqrt(-s*s)*b/2/vD)}
gTR 9 0 laplace {(v(17)-v(18))/Z_0}=
+{-2*h15*sin(sqrt(-s*s)*b/2/vD)/sqrt(-s*s)/Z_0}
.ac lin 10001 1 299.9k
.probe
.end
```

Figure 6 – PSpice listing for the complete equivalent circuit

All geometrical parameters from figure 2 can be identified in the library file KLM.lib. The variable α denotes the ratio between the diameter of the preloading ring and width of the piezoceramic plate. In order to obtain a lower resonance frequency, its value has to be kept as small as possible. On the other hand, if α is too small, the mechanical stresses may cause physical damage near the edges of the thin plate. Finally, a compromise value of 0.33 was chosen for α. The nodes of the circuit presented in figure 5 are the same as the ones used in the listing from figure 6. For simplicity, it was considered that the area of the center base is the same as the area of the inner-most layer from the seismic mass approximation. For preventing too large simulation times, the number of layers was taken $n = 10$ at each side. The listing contains also seven resistors which have no AC functionality but avoiding floating nodes in the circuit. This is the reason why all of them have huge values, of 1 [TΩ].

NUMERICAL RESULTS

A parametric plot was done for the thickness of the piezoceramic between 1 and 5 [mm]. As long as all other geometries are expressed as a function of *b*, the following decaying rule for the resonant frequency was obtained.

$$f_{res} = 15.00377 + 71.71315\, e^{-\frac{b-0.91699e-3}{0.00229}} + 59.85123\, e^{-\frac{b-0.91699e-3}{5.39466e-4}} \text{ [kHz]} \qquad (10)$$

For example, if the thickness *b* is 1 [mm] results a $f_{res} \approx 146.5$ [kHz]. This calculated value is higher than the mounted resonance frequency of a real accelerometer, because in that case the value is lowered by the case-seismic mass-center base pre-loading force. The value is also much smaller when compared to the natural frequency of a 1 [mm] thick PZ-23 shear resonator, which is 946.1 [kHz] [15]. Depending on the accelerometer's mechanical case and on the preloading methods, relatively small transducers can be designed to resonate at frequencies below 5…10 [kHz] [5].

A simulation concerning the KLM impedance of the accelerometer, for illustrating the resonant and anti-resonant frequencies, is presented in figure 7. It can be clearly drawn out that with the increasing of the piezoceramic thickness the first resonant frequency is decreasing [3].



Figure 7 - Logarithmic plots for the KLM impedance as a function of the piezoceramic element width and frequency

SUMMARY

The KLM approach was used for a relatively simple shear-type accelerometer, starting from the equivalent circuit of a single piezoceramic plate. The standard KLM model was extended in order to include the coupling effects between the center base and the plates, and to approximate as close as possible the real shape of the seismic masses. A special attention was given to implement accurately in the frequency domain the equivalent circuit. Finally, an impedance plot for the calculated spectra was presented for showing the resonant and anti-resonant frequencies.

The design principles here presented can be used for simulating in frequency domain other types of accelerometers, when the spectra are of interest during the design phase.

REFERENCES

[1] Arnau, A. (Ed.), *Piezoelectric Transducers and Applications*. (Springer, Berlin, 2004).

[2] Cao, W. *et al*, "Analysis of shear modes in a piezoelectric vibrator", J. Appl. Phys., **83** (8), 4415-4420 (1998).

[3] Harris, C.M., C.E. Crede, *Shock and Vibration Handbook*. (McGraw-Hill, New York, 1961).

[4] Leach, W.M., "Controlled-Source Analogous Circuits and SPICE Models for Piezoelectric Transducers", IEEE Trans. on Ultrasonics, Ferroelectrics, and Frequency Control, **41** (1), 60-66 (1994).

[5] Licht, T.R. *et al*, "Recent Developments in Accelerometer Design", Brüel & Kjær Technical Review, **2**, 1-22 (1987).

[6] Lucklum, R., *Resonante Sensoren*. (in German, Habilitationsschrift, University Otto-von-Guericke, 2002).

[7] Serridge, M., T.R. Licht, *Piezoelectric Accelerometers and Vibration Preamplifiers*. (Brüel & Kjær, Glostrup, 1986).

[8] Sherrit, S. *et al*, "Comparison of the Mason and KLM Equivalent Circuits for Piezoelectric Resonators in the Thickness Mode", IEEE Ultrasonics Symposium, **2**, 921-926 (1999).

[9] Ursu, M., "The KLM Approach for Designing Piezoelectric Transducers", Acta Technica Napocensis, Applied Mathematics and Mechanics Series, **48**, 253-262 (2005).

[10] Vel, S.S., R.C. Batra, "Exact Solution for Rectangular Sandwich Plates with Embedded Piezoelectric Shear Actuators", AIAA Journal, **39** (7), 1363-1373 (2001).

[11] Wang, J., L. Shen, "Exact thickness-shear resonance frequency of electroded piezoelectric crystal plates", J. Zhejiang Univ. SCI., **6A** (9), 980-985 (2005).

[12] Yang, J.S., "Analysis of ceramic thickness shear piezoelectric gyroscopes", J. Acoust. Soc. Am., **102** (6), 3542-3548 (1997).

[13] "MicroSim Application Notes, Version 8.0", MicroSim Corporation (1997).

[14] "PSpice User's Guide, Product Version 10.0", Cadence Design Systems (2003).

[15] http://www.ferroperm-piezo.com/ferroperm_material_data.xls